

Image method for efficiently simulating small-room acoustics

Jont B. Allen and David A. Berkley

Acoustics Research Department, Bell Laboratories, Murray Hill, New Jersey 07974
(Received 6 June 1978)

Image methods are commonly used for the analysis of the acoustic properties of enclosures. In this paper we discuss the theoretical and practical use of image techniques for simulating, on a digital computer, the impulse response between two points in a small rectangular room. The resulting impulse response, when convolved with any desired input signal, such as speech, simulates room reverberation of the input signal. This technique is useful in signal processing or psychoacoustic studies. The entire process is carried out on a digital computer so that a wide range of room parameters can be studied with accurate control over the experimental conditions. A FORTRAN implementation of this model has been included.

PACS numbers: 43.55.Ka, 43.55.Br

INTRODUCTION

In some recent experiments, which studied the perceptual effects of reverberation properties of a small room,^{1,2} a carefully controlled, easily changed, acoustic environment was required. It was decided to utilize a computer simulation of the acoustic space. This paper describes both the general theoretical approach and the specific implementation techniques used (the FORTRAN program). We believe that the resulting room model is useful for a broad range of investigations, from our original experiments mentioned above, to basic studies of room acoustics.

The room model assumed is a rectangular enclosure with a source-to-receiver impulse response, or transfer function, calculated using a time-domain image expansion method. Frequent applications have been made of the image method in the past as in deriving the reverberation-time equations,³ for theoretical studies of sound behavior in enclosures,⁴⁻⁷ and in the study of architectural acoustics and perceptual properties of rooms.⁸⁻¹¹ In addition, there has been a considerable amount of important theoretical work on the approximate¹² use of images produced by a single soft-wall (finite impedance) reflection. Several recent papers on this subject which have good bibliographies are Refs. 13, 14, and 15. Computer methods have also recently been applied to image computations in enclosures (see for example Refs. 6, 7, 10, and 11). In the current paper the computational technique is specifically aimed at being simple, easy to use, and fast. In addition the resulting room responses have been used to realistically model speech transmission in rooms and to investigate the effects of various forms of digital speech signal processors.^{16,17}

In the following we will first briefly discuss theoretical aspects of the method. Then we will outline the computational approach and, finally, we will give some examples of applications.

I. IMAGE VERSUS NORMAL MODE MODELS

We model the rooms of interest as simple rectangular enclosures. This choice of geometry is made for several reasons:

(1) We are most interested in the office environment, which is usually a rectangular geometry.

(2) This model can be most easily realized in an efficient computer program.

(3) The image solution of a rectangular enclosure rapidly approaches an exact solution of the wave equation as the walls of the room become rigid.

The image model is chosen because we are interested in the point-to-point (e.g., talker-to-microphone) transfer function of the room. In order to obtain a good transient description of the response, a time domain model is required. A normal-mode solution of the enclosure would require calculation of all modes within the frequency range of interest (i.e., 0.1–4.0 kHz), plus corrections for those outside this range. The image method includes only those images contributing to the impulse response. Thus the contributing images are those within a radius given by the speed of sound times the reverberation time.⁶ (The exact relationship between the normal-mode solutions and the image solution, for a lossless room, is discussed in Appendix A.) The important information used here is that in the time-domain, each image contributes only a pure impulse of known strength and delay while each normal mode is a decaying exponential which contributes to all times. Furthermore, whereas an image has only delay and gain as parameters, a normal mode computation requires the solution of transcendental equations to find the pole location plus the evaluation of a relatively complex function to find the mode gain (i.e., the residue of the pole).

A. The image model

We model a talker in a room as a point source in a rectangular cavity. A single frequency point source of acceleration in free space emits a pressure wave of the form

$$P(\omega, X, X') = \frac{\exp[i\omega(R/c - t)]}{4\pi R}, \quad (1)$$

where

P = pressure ,

$\omega = 2\pi f$,

f = frequency ,

t = time ,

$R = |\mathbf{X} - \mathbf{X}'|$, (2)

\mathbf{X} = vector talker location (x, y, z) ,

\mathbf{X}' = vector microphone location (x', y', z') ,

$i = \sqrt{-1}$,

c = speed of sound .

When a rigid wall is present, the rigid wall (zero normal velocity) boundary condition may be satisfied by placing an image symmetrically on the far side of the wall. Thus,

$$P(\omega, \mathbf{X}, \mathbf{X}') = \left[\frac{\exp[i(\omega/c)R_-]}{4\pi R_-} + \frac{\exp[i(\omega/c)R_+]}{4\pi R_+} \right] \exp(-i\omega t) , \quad (3)$$

where we define the two distances from the microphone to the source R_- and to the image R_+ by

$$R_-^2 = (x - x')^2 + (y - y')^2 + (z - z')^2 , \quad (4)$$

$$R_+^2 = (x + x')^2 + (y - y')^2 + (z - z')^2 .$$

The wall has been placed at $x=0$ in this case (note the sign in the x terms of R_- and R_+).

In the general case of six walls the situation becomes more complicated because each image is itself imaged. The pressure may then be written (as shown in Appendix A)

$$P(\omega, \mathbf{X}, \mathbf{X}') = \sum_{p=1}^8 \sum_{r=-\infty}^{\infty} \frac{\exp[i(\omega/c)|\mathbf{R}_p + \mathbf{R}_r|]}{4\pi|\mathbf{R}_p + \mathbf{R}_r|} \exp(-i\omega t) , \quad (5)$$

where \mathbf{R}_p represents the eight vectors given by the eight permutations over \pm of

$$\mathbf{R}_p = (x \pm x', y \pm y', z \pm z') \quad (6)$$

r is the integer vector triplet (n, l, m) , and

$$\mathbf{R}_r = 2(nL_x, lL_y, mL_z) , \quad (7)$$


where (L_x, L_y, L_z) are the room dimensions. Equation (5) is the pressure frequency response assuming rigid walls for a point source at $\mathbf{X} = (x, y, z)$ and receiver at $\mathbf{X}' = (x', y', z')$. If Eq. (5) is Fourier transformed, we find the room impulse response function (time domain Green's function)

$$p(t, \mathbf{X}, \mathbf{X}') = \sum_{p=1}^8 \sum_{r=-\infty}^{\infty} \frac{\delta[t - (|\mathbf{R}_p + \mathbf{R}_r|/c)]}{4\pi|\mathbf{R}_p + \mathbf{R}_r|} . \quad (8)$$

An interpretation of Eq. (8) is given in Fig. 1 where we show a part of the image space for a two-dimensional slice through the room. When the accelerative source location (talker) \mathbf{X} is excited, each image point is simultaneously excited, creating spherical pressure waves which propagate away from each image point.

Equation (8) is the exact solution to the wave equation in a rectangular, rigid-wall (lossless), room and may

IMAGE EXPANSION $p(t)$



$$p(t) = \sum_{p=1}^8 \sum_{r=-\infty}^{\infty} \frac{\delta\left(t - \frac{|\vec{R}_p + \vec{R}_r|}{c}\right)}{4\pi|\vec{R}_p + \vec{R}_r|}$$

$$\vec{R}_r = (2nL_x, 2lL_y, 2mL_z)$$

$$\vec{R}_p = (x \pm x', y \pm y', z \pm z')$$

$$\vec{R}_p + \vec{R}_r = (x \pm x' + 2nL_x, y \pm y' + 2lL_y, z \pm z' + 2mL_z)$$

FIG. 1. A slice through the image space showing how the images of the source are spatially arranged. The solid box represents the original room. The actual image space is three dimensional.

be derived directly from the normal-mode solution as shown in Appendix A.

B. Case of nonrigid walls

If the room walls are not rigid, the solution in terms of point images may no longer be exact. A precise statement of the effects of finite impedance walls is presently impossible, since the effects on even a single image are quite complicated.^{13,14,15} Therefore we have continued to assume the approximate point image model even for nonrigid walls. In addition we have assumed an angle independent pressure wall reflection coefficient β . This assumption is equivalent to assuming that the wall impedance is proportional to $\sec(\theta)$, where θ is the angle of incidence of a plane wave with respect to the wall normal. We presently do not understand the exact physical interpretation of the above assumptions. However, we believe that they do not introduce serious problems into the final result under typical conditions. By typical, we mean over the frequency range of 100 Hz-4 kHz, wall reflection coefficients of greater than 0.7, typical office room geometries, and where both source and receiver are not close to the wall. Many, if not all, of the above conditions are probably not critical and could be relaxed. We merely wish to carefully point out the nonexact nature of the results.

The above assumptions result in the Sabine energy absorption coefficient α for a uniform reflection coefficient β on a given wall of the form

$$\alpha = 1 - \beta^2 . \quad (9)$$

Our assumptions are similar to those of geometrical acoustics¹⁸ and are the same as those required for specular angle-independent ray tracing. In current implementations of the model we also do not allow frequency variations in the reflection coefficients. Both the angle dependence and frequency dependence could be included

in our computer program, but only at the expense of significantly complicating and slowing down the computational model.

Introducing the effects of finite, angle independent wall absorption into Eq. (8) leads to the modified room impulse response

$$p(t, \mathbf{X}, \mathbf{X}') = \sum_{\mathbf{p}=0}^1 \sum_{\mathbf{r}=-\infty}^{\infty} \beta_{x1}^{|\mathbf{r}-\mathbf{a}|} \beta_{x2}^{|\mathbf{r}|} \beta_{y1}^{|\mathbf{r}-\mathbf{j}|} \beta_{y2}^{|\mathbf{r}|} \beta_{z1}^{|\mathbf{r}-\mathbf{k}|} \beta_{z2}^{|\mathbf{r}|} \times \frac{\delta[t - (|\mathbf{R}_p + \mathbf{R}_r|/c)]}{4\pi |\mathbf{R}_p + \mathbf{R}_r|}, \quad (10)$$

where \mathbf{R}_p is now expressed in terms of the integer 3-vector $\mathbf{p}=(q, j, k)$ as

$$\mathbf{R}_p = (x - x' + 2qx', y - y' + 2jy', z - z' + 2kz'). \quad (11)$$

\mathbf{R}_p as given by Eq. (6) is similar to that of (11), but is indexed differently from (11). The beta's are the pressure reflection coefficients of the six boundary planes, with the subscript 1 referring to walls adjacent to the coordinate origin (see Fig. 1). Subscript 2 is the opposing wall. Eq. (10) has been derived heuristically from geometrical considerations of Fig. 1. The sum \sum with vector index \mathbf{p} is used to indicate three sums, namely one for each of the three components of $\mathbf{p}=(q, j, k)$. $\mathbf{r}=(n, l, m)$ is a similar sum. Physically these sums are over a three-dimensional lattice of points. For \mathbf{p} there are eight points in the lattice and for \mathbf{r} , the lattice is infinite.

II. IMPLEMENTATION OF THE MODEL

The primary consideration in a computer (sampled data) implementation of Eq. (10) is the method of spatial sampling. In addition, an apparently nonphysical behavior of the model at zero frequency is removed by a low-frequency (0.01 of the sampling frequency) high-pass digital filter.¹⁹

A calculated impulse response is built up as a "histogram" of image pulses received at different time delays. The width of each histogram bin is equal to the time sampling period T initially assumed, which in turn is determined by the highest frequency to be represented. For example, all images with the range $N\Delta R$ to $(N+1)\Delta R$, where $\Delta R=cT$ (T is the sampling period and c the speed of sound), are added together with appropriate amplitude as given by Eq. (10).

The choice of sampling rate is governed by the application. If speech is to be studied in small rooms one might choose $T=0.1$ ms. (sampling frequency of 10 kHz; highest frequency of 5 kHz). But, if reverberation times of large enclosures are being studied (and convolution with speech is not required) much lower rates can be useful.

The time length of the calculated impulse response is also a consideration. For a given sampling rate the number of points in $p(t)$ increases linearly with its length while the computation time (and number of images) goes up approximately as the cube of response length. This is shown in the first four columns of Table

TABLE I. Computation parameters—room size (feet) $10' \times 15' \times 12.5'$ 8 kHz sampling rate.

Length (ms)	Impulse response		Computation time (s)	Convolution rate (s/s)
	No. points	Image count		
64	512	585	1	12.5
128	1024	4690	8	13.8
256	2048	37500	60	15.0

I for our implementation on a Data General, *Eclipse* S/200 computer. (On this machine the computation time required for each image is about 1.6 ms.) The actual FORTRAN programs used are given in Appendix B.

The temporal quantization in the impulse function computation causes slight statistical errors in the computed arrival times of each image pulse relative to the exact delay as given in Eq. (10). This error can be thought of as effectively "moving" each image source by $0 \leq \text{error} \leq \Delta R/2$ relative to the receiver. This effect could be removed, in principle, by using a band-limited source pulse. However, the error is small for most (if not all) purposes and it greatly complicates the computation to remove this approximation. We have estimated that the error due to the slight moving of the images could not be perceived even in a digital simulation of a binaural hearing experiment.

The subroutine SROOM of Appendix B requires as parameters the number of impulse response points desired (NPTS), the source location \mathbf{R}_0 , the receiver lo-




FIG. 2. Plot of a typical impulse response for a room $80 \times 120 \times 100$ sample lengths long. Wall reflection coefficients were all 0.9 ceiling and floor coefficients were 0.7. \mathbf{X} and \mathbf{X}' were at (30, 100, 40) and (50, 10, 60) sample periods.

cation R , the room dimensions, all specified in terms of the sample length (ΔR), and the reflection coefficients of each of the six wall surfaces (β). Figure 2 shows an example of the impulse response obtained for a room of dimensions $80 \times 120 \times 100$ sample lengths with equal wall reflection coefficients of 0.9 ($\alpha = 0.19$) and with floor and ceiling reflection coefficients (β_x) of 0.7 ($\alpha = 0.51$). X and X' were (30, 100, 40) and (50, 10, 60) sample intervals, respectively.

It is usually convenient to interpret the model parameters as a true distance rather than as multiples of ΔR . This requires the choice of a sampling rate and then conversions may be performed in the users main program which calls the subroutines of Appendix B. Figure 2 is labeled assuming an 8 kHz sampling rate. For this assumption (and assuming a sound speed of 1 ft/ms) the room dimensions are $10' \times 15' \times 12.5'$.

III. APPLICATIONS

Our room image model has been applied to several problems. We will discuss two examples: a psychophysical evaluation of room reverberation effects^{1,2} and a study of critical distance measurements using spectral response variance.²⁰ We have also used the model to test a signal processor intended to reduce perceived reverberation¹⁶ and to study problems associated with mathematical inversion (inverse filtering) of room transfer functions.¹⁷

A. Psychophysics of room reverberation

Once a simulated room impulse response has been calculated using the image model, the psychophysical effects of this simulated reverberation on speech may be directly studied. A reverberant sample of speech was produced by convolving an anechoic (unreverberant) speech sample with the calculated impulse response [$p(t)$]. This can be done efficiently using a Fast Fourier Transform (FFT) method (overlap-add) to perform the convolution.²¹ The last column of Table I shows the measured convolution rate, for various length impulse responses. The convolution rate only increases as $\log_2(N)$, (where N is the room response length in time sample periods T) so even large impulse responses can be convolved with speech quite efficiently. For example, to convolve (filter) one second of speech, sampled at 8 kHz, with a 256 ms long impulse response (2048 points) requires a 15 s computation.

The speed of processing makes multivariate psychophysical studies quite practical. Ease of modification and perfect control of room parameters avoids the problems which have made such experiments so difficult in the past. The actual experiments used 16 different simulated "rooms" (impulse responses) convolved with ten different sentences spoken by four different speakers. It was discovered that the experimental rooms were perceptually well characterized by their spectral variance [Eq. (14)] and by the reverberation time. This latter measure, reverberation time, deserves some discussion.

Given the impulse responses, reverberation time may be estimated in a number of ways. One method is to

use a modification of the integrated tone-burst method²²

$$E(t) = k \int_t^\infty p^2(\tau) d\tau, \quad (12)$$

where $E(t)$ is the average energy decay, k is a proportionality constant, and $p(\tau)$ is the calculated pressure impulse response from Eq. (10). For cases where the impulse response has been truncated before most of the decay has taken place, (12) may lead to errors. These errors are usually obvious in the $E(t)$ plots.

Another, approximate, approach is to simply measure the short-time average energy decay of the impulse itself (e.g., using a simulated level recorder). For exponential or near-exponential decays, both methods should give approximately the same value of




FIG. 3. (a) Energy decay curve for the impulse response of Fig. 2 using the Schroeder integration method.²² (b) Impulse energy decay curve for a simulated level recorder.

reverberation time. Example plots of $E(t)$ for both methods are shown in Figs. 3(a) and 3(b) using the impulse response of Fig. 2. Experience indicates that Eq. (12) gives the most satisfactory results. We have found

empirically that calculated reverberation times, for a number of simulated enclosures, agree well with Eyring's formula³ over a wide range of Beta values.²³

In the experiments discussed above^{1,24} we discovered a monotonic relationship between Δ/Δ_c (Fig. 4) the microphone-talker distance when normalized by the room critical distance (the distance at which reverberant energy equals direct sound energy), the reverberation time, and psychophysical preference for the resulting speech.

B. Critical distance measurement

A new method has been proposed²⁰ for measurement of critical distances (or reverberation radius) in rooms. In this technique a measurement is made of the log frequency response variance σ_L defined as

$$L(\omega) = 20 \log [|P(\omega)|] \quad (13)$$

$$\sigma_L^2 = \overline{[L(\omega) - \overline{L(\omega)}]^2}, \quad (14)$$

given the room transfer-function $P(\omega)$ [Fourier transform of Eq. (10)] for several microphone-source spacings. The measured values are fitted to a theoretical curve for σ_L based on the assumption of simultaneously excited, uncorrelated, normal modes, combined with the calculated direct sound energy. The resulting fit was shown to give an accurate value for the room's critical distance.

This new method was extensively studied using our image model before being applied successfully to real rooms. Since the direct and reverberant energy are known in the computer model, a comparison can easily be made to the theory. The model results show excellent agreement with theoretical calculation as is seen in Figs. 4(a) and 4(b). We know of no other method by which this study could have been carried out as effectively.

IV. SUMMARY AND DISCUSSION

A simulation method for small rooms based on an approximate image expansion for rectangular nonrigid-wall enclosures has been discussed. The method is simple, easy to implement and efficient for computer simulation. Several examples of its use, where other methods would be difficult, have been discussed.


APPENDIX A

We wish to derive the rigid-wall image solution directly from the normal-mode expansion for a rectangular enclosure. The frequency response function (Green's function) for the pressure $P(\omega)$ in an enclosure is given by solving the Helmholtz equation driven by a single frequency point acceleration source.

$$\nabla^2 P[(\omega/c), \mathbf{X}, \mathbf{X}'] + (\omega^2/c^2)P[(\omega/c), \mathbf{X}, \mathbf{X}'] = -\delta(\mathbf{X} - \mathbf{X}'), \quad (A1)$$

where ω is the frequency and c is the speed of sound. The solution to this equation, assuming rigid boundaries, is given by

(a) STANDARD DEVIATION OF SPECTRAL RESPONSE IN COMPUTER-SIMULATED ROOM: 17' x 13' x 10'



(b) STANDARD DEVIATION OF SPECTRAL RESPONSE IN COMPUTER-SIMULATED ROOM: 47' x 31' x 15'




FIG. 4. Figures from Jetzt²⁰ which compare the theoretical rms deviation of the pressure in dB from the mean pressure in dB as a function of the direct to reverberant energy ratio (a) for a room 17 x 13 x 10 ft and (b) 47 x 31 x 15 ft.

$$P(k, \mathbf{X}, \mathbf{X}') = \frac{1}{V} \sum_{\mathbf{r}=-\infty}^{\infty} \frac{\psi_{\mathbf{r}}(\mathbf{X}) \psi_{\mathbf{r}}(\mathbf{X}')}{(k_{\mathbf{r}}^2 - k^2)}, \quad (\text{A2})$$

where $k = \omega/c$, $\mathbf{r} = (n, l, m)$ indicates a three dimensional sum, V is the room volume,

$$\mathbf{k}_{\mathbf{r}} = \left(\frac{n\pi}{L_x}, \frac{l\pi}{L_y}, \frac{m\pi}{L_z} \right), \quad (\text{A3})$$

$$k_{\mathbf{r}}^2 = |\mathbf{k}_{\mathbf{r}}|^2$$

and

$$\psi_{\mathbf{r}}(\mathbf{X}) = \cos\left(\frac{n\pi x}{L_x}\right) \cos\left(\frac{l\pi y}{L_y}\right) \cos\left(\frac{m\pi z}{L_z}\right). \quad (\text{A4})$$

where the L_i 's are the room dimensions.

Using the exponential expansion for cosine, multiplying the terms of Eq. (A2) together and collecting, we obtain

$$P(k, \mathbf{X}, \mathbf{X}') = \frac{1}{8V} \sum_{\mathbf{r}=-\infty}^{\infty} \sum_{\mathbf{p}=1}^8 \frac{\exp(i\mathbf{k}_{\mathbf{r}} \cdot \mathbf{R}_{\mathbf{p}})}{(k_{\mathbf{r}}^2 - k^2)}, \quad (\text{A5})$$

where $\mathbf{R}_{\mathbf{p}}$ represents the eight vectors [also given by Eq. (6)]

$$\mathbf{R}_{\mathbf{p}} = (x \pm x', y \pm y', z \pm z'). \quad (\text{A6})$$

Using the property of the delta function on k_x , k_y , and k_z

$$\int_{-\infty}^{\infty} \delta(x-a)F(x)dx = F(a), \quad (\text{A7})$$

we may rewrite Eq. (A5) in integral form

$$P(k, \mathbf{X}, \mathbf{X}') = \frac{1}{8V} \sum_{\mathbf{p}=1}^8 \int \int \int \frac{\exp(i\xi \cdot \mathbf{R}_{\mathbf{p}})}{(|\xi|^2 - k^2)} \sum_{\mathbf{r}=-\infty}^{\infty} \delta(\xi - \mathbf{k}_{\mathbf{r}}) d^3 \xi. \quad (\text{A8})$$

By Fourier series analysis one may show

$$\sum_{n=-\infty}^{\infty} \delta\left(\xi_x - \frac{n\pi}{L_x}\right) = \frac{L_x}{\pi} \sum_{n=-\infty}^{\infty} \exp(i2L_x n \xi_x). \quad (\text{A9})$$

Thus [with analogous equations to (A9) for y and z]

$$P(k, \mathbf{X}, \mathbf{X}') = \frac{1}{(2\pi)^3} \sum_{\mathbf{p}=1}^8 \int \int \int \sum_{\mathbf{r}=-\infty}^{\infty} \frac{\exp[ii\xi \cdot (\mathbf{R}_{\mathbf{p}} + \mathbf{R}_{\mathbf{r}})]}{(|\xi|^2 - k^2)} d^3 \xi, \quad (\text{A10})$$

where $\mathbf{R}_{\mathbf{r}}$ is the vector [also given by Eq. (7)]

$$\mathbf{R}_{\mathbf{r}} = 2(nL_x, lL_y, mL_z). \quad (\text{A11})$$

Each triple integral is just a plane wave expansion for a point source in free space since

$$\frac{\exp(ik|\mathbf{R}|)}{4\pi|\mathbf{R}|} = \frac{1}{8\pi^3} \int \int \int \frac{\exp(i\xi \cdot \mathbf{R})}{(|\xi|^2 - k^2)} d^3 \xi. \quad (\text{A12})$$

Finally, using Eq. (A12), Eq. (A10) becomes

$$P\left(\frac{\omega}{c}, \mathbf{X}, \mathbf{X}'\right) = \sum_{\mathbf{p}=1}^8 \sum_{\mathbf{r}=-\infty}^{\infty} \frac{\exp[i(\omega/c)|\mathbf{R}_{\mathbf{p}} + \mathbf{R}_{\mathbf{r}}|]}{4\pi|\mathbf{R}_{\mathbf{p}} + \mathbf{R}_{\mathbf{r}}|}. \quad (\text{A13})$$

Taking the inverse Fourier transform of Eq. (A13), the echo structure becomes explicit

$$p(t, \mathbf{X}, \mathbf{X}') = \sum_{\mathbf{p}=1}^8 \sum_{\mathbf{r}=-\infty}^{\infty} \frac{\delta[t - (|\mathbf{R}_{\mathbf{p}} + \mathbf{R}_{\mathbf{r}}|/c)]}{4\pi|\mathbf{R}_{\mathbf{p}} + \mathbf{R}_{\mathbf{r}}|}, \quad (\text{A14})$$

which is the same as Eq. (8) as desired.

APPENDIX B

```

C PGM: SROOM
C SUBROUTINE TO CALCULATE A ROOM IMPULSE RESPONSE
C R=VECTOR RADIUS TO RECEIVER IN SAMPLE PERIODS=LENGTH/(C*T)
C R0=VECTOR RADIUS TO SOURCE IN SAMPLE PERIODS
C RL=VECTOR OF BOX DIMENSIONS IN SAMPLE PERIODS
C BETA=VECTOR OF SIX WALL REFLECTION COEFS (0 < BETA <=1)
C HT=IMPULSE RESP ARRAY
C NPTS=# OF POINTS OF HT TO BE COMPUTED
C ZERO DELAY IS IN HT(1)
C
SUBROUTINE SROOM(R, R0, RL, BETA, HT, NPTS)
  DIMENSION HT(NPTS)
  DIMENSION R(3), R0(3), NR(3), RL(3), DELP(8), BETA(2,3)
  EQUIVALENCE (NR(1), NX), (NR(2), NY), (NR(3), NZ)
  DO 5 I=1, NPTS
5    HT(I)=0
C CK FOR MIC AND SOURCE AT SAME LOCATION
  DIS=0
  DO 6 I=1,3
6    DIS=(R(I)-R0(I))**2+DIS
  DIS=SQRT(DIS)
  IF(DIS.LT..5)HT(1)=1
  IF(DIS.LT..5)RETURN
C FIND RANGE OF SUM
  N1=NPTS/(RL(1)*2)+1
  N2=NPTS/(RL(2)*2)+1
  N3=NPTS/(RL(3)*2)+1
  DO 20 NX=-N1, N1
  DO 20 NY=-N2, N2
  DO 20 NZ=-N3, N3

```

```

C GET EIGHT IMAGE LOCATIONS FOR MODE # NR
  CALL LTHIMAGE(R, R0, RL, NR, DELP)
  I0=0
  DO 10 L=0,1
  DO 10 J=0,1
  DO 10 K=0,1
  I0=I0+1
C MAKE DELAY AN INTEGER
  ID=DELP (I0) + .5
  FDM1=ID
  ID=ID+1
  IF (ID.GT.NPTS)GO TO 10
C PUT IN LOSS FACTOR ONCE FOR EACH WALL REFLECTION
  GID=BETA(1,1)**IABS(NX-L)
  1 *BETA(2,1)**IABS(NX)
  2 *BETA(1,2)**IABS(NY-J)
  3 *BETA(2,2)**IABS(NY)
  4 *BETA(1,3)**IABS(NZ-K)
  5 *BETA(2,3)**IABS(NZ)
  6 /FDM1
C CHECK FOR FLOATING POINT UNDERFLOW HERE;
C IF UNDER FLOW, SKIP NEXT LINE
  HT(ID)=HT(ID)+GID
10 CONTINUE
20 CONTINUE
C IMPULSE RESP HAS BEEN COMPUTED
C FILTER WITH HI PASS FILT OF 1% OF SAMPLING FREQ (I.E. 100 HZ)
C IF THIS STEP IS NOT DESIRED, RETURN HERE
  W=2.*4.*ATAN(1.)*100.
  T=1E-4
  R1=EXP(-W*T)
  R2=R1
  B1=2.*R1*COS(W*T)
  B2=-R1*R1
  A1=-(1.+R2)
  A2=R2
  Y1=0
  Y2=0
  Y0=0
C FILTER HT
  DO 40 I=1, NPTS
  X0=HT(I)
  HT(I)=Y0+A1*Y1+A2*Y2
  Y2=Y1
  Y1=Y0
  Y0=B1*Y1+B2*Y2+X0
40 CONTINUE
  RETURN
  END

```

} NOTE CONTINUATION
LINES

```

C PGM: LTHIMAGE
C PGM TO COMPUTE EIGHT IMAGES OF A POINT IN BOX
C
  SUBROUTINE LTHIMAGE(DR, DR0, RL, NR, DELP)
C
C DR IS VECTOR RADIUS TO RECEIVER IN SAMPLE PERIODS
C DR0 IS VECTOR RADIUS TO SOURCE IN SAMPLE PERIODS
C RL IS VECTOR OF BOX DIMENSIONS IN SAMPLE PERIODS
C NR IS VECTOR OF MEAN IMAGE NUMBER
C DELP IS VECTOR OF EIGHT SOURCE TO IMAGE
C DISTANCES IN SAMPLE PERIODS
C
  DIMENSION R2L(3), RL(3), NR(3),DELP( 8)
  DIMENSION DR0(3), DR(3), RP(3,8)
C LOOP OVER ALL SIGN PERMUTATIONS AND COMPUTE R+/-R0
  I0=1
  DO 10 L=-1,1,2
  DO 10 J=-1,1,2
  DO 10 K=-1,1,2

```

```

C NEAREST IMAGE IS L=J=K=-1
  RP (1, I0) = DR(1) + L*DR0(1)
  RP(2, I0) = DR(2) + J*DR0(2)
  RP(3, I0) = DR(3) + K*DR0(3)
  I0 = I0 + 1
10 CONTINUE
C ADD IN MEAN RADIUS TO EIGHT VECTORS TO GET TOTAL DELAY
  R2L(1) = 2.*RL(1)*NR(1)
  R2L(2) = 2.*RL(2)*NR(2)
  R2L(3) = 2.*RL(3)*NR(3)
  DO 20 I=1,8
  DELSQ = 0
  DO 25 J=1,3
  R1 = R2L(J) - RP(J,I)
  DELSQ = DELSQ + R1**2
25 CONTINUE
  DELP(I) = SQRT(DELSQ)
20 CONTINUE
  RETURN
  END

```

¹Barbara McDermott and Jont Allen, "Perceptual Factors of Small Room Reverberation," *J. Acoust. Soc. Am.* 60, S9 (A) (1976).

²Jont B. Allen and B. J. McDermott, "A New Method for Measuring Perception of Room Reverberation" (unpublished).

³C. F. Eyring, "Reverberation Time in 'Dead' Rooms," *J. Acoust. Soc. Am.* 1, 217-241 (1930).

⁴D. Mintzer, "Transient Sounds in Rooms," *J. Acoust. Soc. Am.* 22, 341-352 (1950).

⁵R. H. Bolt, P. E. Doak, and P. J. Westervelt, "Pulse Statistics Analysis of Room Acoustics," *J. Acoust. Soc. Am.* 22, 328-339 (1950).

⁶J. M. Berman, "Behavior of Sound in a Bounded Space," *J. Acoust. Soc. Am.* 57, 1275-1291 (1975).

⁷E. K. Dunens and R. F. Lambert, "Impulsive Sound Response Statistics," *J. Acoust. Soc. Am.* 61, 1524-1532 (1977).

⁸J. R. Power, "Measurement of Absorption in Rooms with Sound Absorbing Ceilings," *J. Acoust. Soc. Am.* 10, 98-101 (1938).

⁹C. G. Mayo, "Standing Wave Patterns in Studio Acoustics," *Acustica* 2, 49-64 (1951).

¹⁰B. M. Gibbs and D. K. Jones, "A Simple Image Method for Calculating the Distribution of Sound Pressure Levels within an Enclosure," *Acustica* 26, 24-32 (1972).

¹¹F. Santon, "Numerical Prediction of Echograms and the Intelligibility of Speech in Rooms," *J. Acoust. Soc. Am.* 59, 1399-1405 (1976).

¹²By "approximate" we mean the blurring of the image as a result of the finite impedance wall. Variations in wall impedance as a function of frequency or other frequency dispersion effects will also make the images frequency dependent. In practice the two effects usually appear simultaneously.

¹³A. R. Wentzel, "Propagation of Waves Along an Impedance Boundary," *J. Acoust. Soc. Am.* 55, 956-963 (1974).

¹⁴S. Thomasson, "Reflection of Waves from a Point Source by

an Impedance Boundary," *J. Acoust. Soc. Am.* 59, 780-785 (1976).

¹⁵R. J. Donato, "Spherical Wave Reflections from a Boundary of Reactive Impedance using a Modification of Cagniard's Methods," *J. Acoust. Soc. Am.* 60, 999-1002 (1976).

¹⁶Jont B. Allen, D. A. Berkley, and J. Bauert, "Multimicrophone signal-processor technique to remove room reverberation from Speech Signals," *J. Acoust. Soc. Am.* 62, 912-915 (1977).

¹⁷S. T. Neeley and Jont B. Allen, "Invertibility of a Room Impulse Response," *J. Acoust. Soc. Am.* (in press).

¹⁸P. M. Morse and K. U. Ingard, *Theoretical Acoustics* (McGraw-Hill, New York, 1968), Sec. 9.5.

¹⁹Without the high-pass filter, the Fourier transform of $p(t)$ shows a large energy spike at zero frequency as the absorption α goes to zero. This is in part a direct result of the no-pressure-release boundary conditions we have assumed and is further complicated by the assumption of an accelerative source, which is nonphysical for very low frequencies (less than 50 Hz).

²⁰J. J. Jetzt, "Critical Distance Measurement of Rooms from the Sound Energy Spectral Response," *J. Acoust. Soc. Am.* (to be published).

²¹Jont B. Allen, "FASTFILT—An FFT Based Filtering Program," in *IEEE Press book on Programs for Digital Signal Processing* (IEEE Press, New York, 1979).

²²M. R. Schroeder, "New Method of Measuring Reverberation Time," *J. Acoust. Soc. Am.* 37, 409-412 (1965).

²³Agreement is best for beta the same on all surfaces and variations appear when a pair of opposing walls are significantly different in reflectivity than all other surfaces, (as is the case for Figs. 2 and 3).

²⁴D. A. Berkley, "Normal Listeners in Typical Rooms: Reverberation Perception, Simulation and Reduction" (unpublished).